

257 / 757

Programming in the Arts with Processing

In Class Exercise #7**Day:****Thur 2.6.14****Due:****In Class****Goal: Rotation, Translation, and
Control in 3D****Course Web Site:**<http://www.andrew.cmu.edu/course/60-257/>**Reading:**

Posted on the calendar web page available from the link shown above.

Assignment:

Use the code below and on the next page as your starting code.

You need to add two functions to take user input:

- **mouseMoved()**
which will alter the fill value of the box:
 - increase the value of **redValue** if the mouse moves to the right
 - decrease the value of **redValue** if the mouse moves to the left
- **keyPressed()**
which will react to four key inputs:
 - '+' will increase the value of **rotXDelta** by 1
 - '-' will decrease the value of **rotXDelta** by 1
 - ' ' will increase the value of **transZDelta** by 1
 - 'r' will reset all variables to their original value (DO NOT call **setup()**)

Starting Code:

```
float rotX, rotXDelta;
float redValue;

float transZ, transZDelta;

void setup ( )
{
  size( 800, 500, P3D );
  rotXDelta = 0;
  rotX = 0;
  redValue = 127;

  transZ = 0;
  transZDelta = 5;

  textSize( 18 );
  textAlign( CENTER );

  rectMode( CENTER );
}
```

```
void draw ( )
{
  pushMatrix( );
  prepareScreen( );
  drawFigure( );
  popMatrix( );
  showStats( );
}

void drawFigure( )
{
  noStroke( );
  fill( redValue, 0, 0 );
  box( 100 );
  fill( 255, 255, 0, 70 );
  rect( 0, 0, 40, 30 );
  pushMatrix( );
  fill( 0, 200, 200, 70 );
  rotateY(radians( frameCount ) );
  box( 50 );
  popMatrix( );
  pushMatrix( );
  fill( 200, 0, 200, 70 );
  rotateY(radians( -frameCount*2 ) );
  box( 20 );
  popMatrix( );
}

void prepareScreen( )
{
  background( 0 );
  lights( );
  translate( width/2, height/2, transZ );
  stroke( 200, 200, 0 );
  line( -200, 0, 0, 200, 0, 0 );
  fill( 200, 200, 0 );
  text( "X axis before rotation.", -200, -5, 0 );
  rotX += rotXDelta;
  rotateX( radians( 20 ) );
  rotateY( radians( rotX ) );
  text( "X axis during rotation.", 200, -5, 0 );
  line( -200, 0, 0, 200, 0, 0 );
}

void showStats( )
{
  fill( 255, 0, 0 );
  text( "redValue = " + redValue, 100, 20 );
  fill( 0, 255, 0 );
  text( "rotX = " + rotX, 100, height - 40 );
  text( "rotXDelta = " + rotXDelta, 100, height - 20 );
  text( "transZ = " + transZ, 100, height-60 );
}
```

